you-try-it-02.xlsx
Step-by-Step Guide
ver. 7/26/2016

Abstract

This document provides step-by-step instructions for the Excel workbook you-try-it-02.xlsx (saved from Excel 2010). The worksheets contain data for practice exercises keyed to Chapter 2 of:

Basics of Analytical Chemistry and Chemical Equilibria by Brian M. Tissue
(John Wiley: New York, 2013).

Worksheets in the workbook **page in this guide**

2.A sample-prep 2
2.B single-extraction 6
2.C multiple-extractions 9
2.D extractant-volume 10
2.E percent-recovery (with internal standard calibration) 13

General Advice

It is often useful to work out the first exercise of each worksheet on paper. After obtaining a result, compare your result to the answer in the worksheet. Next try writing formulas to do the calculation(s). If you do not get the same answer, try writing formulas stepwise and check intermediate steps to error check your work. Try to write formulas to be general so that you can copy them to use for multiple cases.

Version History

7/6/2009	First posting.
8/26/2009	Minor errors and formatting corrected.
7/26/2016	Formatting revisions.

For updates visit the text support website:

http://www.achem.org

2.A sample-prep

This worksheet contains two sample preparation procedures for the analysis of lipid-soluble vitamins in vitamin tablets. The sample preparation procedure isolates the analyte from interferences and leaves it in a suitable solution for subsequent measurement. The task is to calculate the concentration of each analyte in the solution to be analyzed based on the amount in the starting sample. We will do the calculation assuming that there is no loss of analyte in any of the preparation steps.

Concepts	**Concept Synopsis**
sample prep calculations	Given an expected amount of analyte in a sample, we must often determine if the sample preparation procedure will result in an analyte concentration that is in the working range of our measurement. This calculation lets us predict the amount of sample that we will need and if concentrating steps might be necessary.

Most of the work in this worksheet is to translate the sample prep steps to conversion factors. This exercise can be completed as easily on paper as with Excel.

1. Calculate the ppm concentrations of vitamins A, D, E and K in the final extract solution for Procedure 2.A.1.

The grinding of the 20 tablets is to ensure homogeneity in the analyzed sample. We start with some amount of this powdered sample (0.6 g) and process it to isolate the analytes. Since we know the expected analyte concentration in each 1.5 g tablet, we can predict the absolute amount of each analyte in the 0.6 g of powder. The analyte concentrations for vitamins A, D, and E are given as I.U. per tablet, so there is an extra unit conversion to convert these analytes to mass. Given analyte amount, the analyte concentration is simply the amount per final volume. The last calculational step is to report the concentrations in the requested units. If we choose appropriate units for our intermediate calculations, we won't need this unit conversion step at the end. Overall our calculation is:

start: amount of sample (0.6 g)
↓ (amount of sample)(analyte concentration)
amount of analyte
↓ (amount of analyte) / (volume of solvent)

concentration of analyte

↓ (unit conversion)

end: analyte concentration in ppm (µg/mL)

First we must get the values that we need for the calculations out of the text of the procedure and into cells that we can reference. I enter the following labels in cells B38:B40:

sample mass:

tablet mass:

extract volume:

In cells C38:C40 I enter the values and in cells D38:D40 I list the appropriate units. After adjusting formatting, this information looks like:

	sample mass:	0.60	g
	tablet mass:	1.5	g
	extract volume:	50.0	mL

Now we can determine the intermediate quantities that were outlined above. For vitamin A, the conversion of analyte concentration from 'I.U. per tablet' to 'I.U. per gram of sample' is:

$$\frac{5000 \text{ I.U. vitamin A}}{1 \text{ tablet}} \quad \frac{1 \text{ tablet}}{1.5 \text{ g}} = \frac{3333 \text{ I.U. vitamin A}}{1.0 \text{ g sample}}$$

I type labels to create a table in cells B42:G47 (see figure below). For vitamin A I enter in cell C44:

=D16/$C$39

Now multiplying this concentration by the amount of sample gives us the amount of analyte:

$$\frac{3333 \text{ I.U. vitamin A}}{1.0 \text{ g sample}} \text{ (0.60 g sample) } = \text{ 2000 I.U. vitamin A}$$

In cell D44 I enter:

=C44*$C$38

Now to convert units from 'I.U. vitamin A' to 'µg vitamin A' I multiply by the conversion factor in Table 2.A:

$$(2000 \text{ I.U. vitamin A}) \frac{0.30 \text{ µg vitamin A}}{1.0 \text{ I.U. vitamin A}} = 600 \text{ µg vitamin A}$$

In cell E44 I enter:

=D44*F16

Finally to determine the concentration, I divide the amount of vitamin A by the final extract volume (50.0 mL).

$$\frac{600 \text{ µg vitamin A}}{50.0 \text{ mL}} = 12 \text{ µg /mL}$$

In cell F44 I type:

=E44/$C$40

Since my concentration is µg/mL, I add the units ppm in cell G44.

The results are in the following figure. I highlight cells C28:G28 and double click on the small square handle at the lower right of the highlighted cells to copy the formulas down the table for the other vitamins. I'll get an error message for vitamin K, because there is no I.U. to µg conversion in Table 2.A.

	A	B	C	D	E	F	G
37							
38		sample mass:	0.60	g			
39		tablet mass:	1.5	g			
40		extract volume:	50.0	mL			
41							
42			analyte	analyte	analyte	analyte	
43		analyte	per g	amount	µg	concentration	
44		vitamin A	3333	2000	600	12	ppm
45		vitamin D					ppm
46		vitamin E					ppm
47		vitamin K					ppm

In cell E47 I type the following formula to fix the error:

=D47

Now the results for Procedure 2.A.1 will look like:

	A	B	C	D	E	F	G
41							
42			analyte	analyte	analyte	analyte	
43		analyte	per g	amount	µg	concentration	
44		vitamin A	3333	2000	600	12	ppm
45		vitamin D	267	160	4	0.08	ppm
46		vitamin E	20	12	12000	240	ppm
47		vitamin K	17	10	10	0.2	ppm

2. Calculate the ppm concentrations of vitamins A and E in the final extract solution for Procedure 2.A.2.

In principle this calculation is the same as above. One difference compared to Procedure 2.A.1 is that after obtaining the 50.0 mL of extract, a 10.0 mL portion is removed for analysis. We are taking 1/5 of the extract for analysis, which saves the remainder for other measurements. We will include this factor of 5 in the calculation of the analyte amount.

In cells B72:D74 I type the labels and constant information as I did for Procedure 2.A.1. The sample mass is now 2.0 g, the tablet mass is still 1.5 g, and the final extract volume is 1.0 mL. I add an extra label for 'extract fraction:' in cell B75 and enter the factor of 1/5 in cell C75 for the fraction of sample that is measured. The one difference that I make is to include the factor of 5 in finding the analyte µg.

I complete the calculations as in Procedure 2.A.1 by typing the equivalent formulas. To include the factor of 1/5 in the sample procedure, in cell E79 I type the formula:

=D79*F16*$C$75

I can't simply copy the formulas down a row because I don't have the same sequence as in Table 2.A. In cell E80 for vitamin E I type:

=D80*F18*$C$75

Since the final volume is 1.0 mL, the final concentrations are numerically the same as the analyte amounts. The final results will look something like the following figure:

	A	B	C	D	E	F	G
71							
72		sample mass:	2.00	g			
73		tablet mass:	1.5	g			
74		extract volume:	1.0	mL			
75		extract fraction:	0.2				
76							
77			analyte	analyte	analyte	analyte	
78		analyte	per g	amount	µg	concentration	
79		vitamin A	3333	6667	400	400	ppm
80		vitamin E	20	40	8000	8000	ppm

2.B single-extraction

This worksheet contains a table with octanol-water log*P* values for various solutes. The exercise is to determine the fraction of solute remaining in the aqueous phase and the fraction transferred to the organic phase for each of the solutes.

Concepts	**Concept Synopsis**
liquid-liquid extraction	A solute will partition between two immiscible phases depending on its relative solubility in the two solvents.
extraction efficiency	A calculation to predict the fraction of a solute remaining in the aqueous phase when extracted with an immiscible organic solvent. The expression is given in exercise 2 below.

1. Write a formula to convert the logP values to K_D'.

The calculation is the inverse log: $K_D' = 10^P$. For the log*P* in cell C16, type in cell D16 the Excel formula:

=10^C16

Copy this formula down the column in Table 2.B.1 and also copy to column D in Table 2.B.2. The results, after adjusting the formatting for the values is shown in the next figure. I have also typed and copied the volumes of the aqueous and organic phases and the value of *n* for the extraction.

	A	B	C	D	E	F	G	H	I
13									
14		**Table 2.B.1**							
15		compound	log*P*	K_D'	V_{aq} (mL)	V_{org} (mL)	*n*	$\alpha_{S(aq)}$	$\alpha_{S(org)}$
16		1-hexanol	2.03	1.1E+02	10	10	1		
17		benzene	2.13	1.3E+02	10	10	1		
18		1-octanol	3.07	1.2E+03	10	10	1		
19		cyclohexane	3.44	2.8E+03	10	10	1		
20		hexane	4.00	1.0E+04	10	10	1		
21		octane	5.15	1.4E+05	10	10	1		
22									
23									
24		**Table 2.B.2**							
25		compound	log*P*	K_D'	V_{aq} (mL)	V_{org} (mL)	*n*	$\alpha_{S(aq)}$	$\alpha_{S(org)}$
26		methanol	-0.74	0.18	10	10	1		
27		ethanol	-0.30	0.50	10	10	1		
28		acetone	-0.24	0.58	10	10	1		
29		acetic acid	-0.17	0.68	10	10	1		

2. Write a formula to find the fraction remaining in the aqueous phase, , $\alpha_{S(aq)}$.

The formula to determine the fraction of the solute remaining in the aqueous phase, $\alpha_{S(aq)}$, is:

$$\alpha_{S(aq)} = \left(\frac{1}{1 + K_D'(V_{org}/V_{aq})} \right)^{n}$$

where the terms in the expression are:

- K_D' distribution constant
- V_{aq} volume of the aqueous phase
- V_{org} volume of the organic phase
- *n* number of sequential extractions

The Excel formula for this expression for 1-hexanol is:

=(1/(1+D16*F16/E16))^G16

Type this formula in cell H16 and copy it down the column for Table 1. Also copy it to column H in Table 2.

You can also use extraction.xlsx to do these calculations and copy the results here. The formula is simple enough that it is easier to type it in rather than copying results from one spreadsheet to another.

3. Convert $\alpha_{S(aq)}$ to $\alpha_{S(org)}$.

Since we are not changing the solute (we hope), the total amount of solute does not change. The sum of the fraction of the solute remaining in the aqueous phase plus the fraction of solute transferred to the organic phase must equal one:

$$\alpha_{S(aq)} + \alpha_{S(org)} = 1$$

The Excel formula for this expression for $\alpha_{S(aq)}$ of 1-hexanol is:

=1-H16

Typing this formula in cell I16 and copying it down the column for Tables 2.B.1 and 2.B.2 gives our final results. The next figure shows the results for all of the calculations.

	A	B	C	D	E	F	G	H	I
13									
14		**Table 2.B.1**							
15		compound	log*P*	K_D'	V_{aq} (mL)	V_{org} (mL)	*n*	$\alpha_{S(aq)}$	$\alpha_{S(org)}$
16		1-hexanol	2.03	1.1E+02	10	10	1	0.009	0.991
17		benzene	2.13	1.3E+02	10	10	1	0.007	0.993
18		1-octanol	3.07	1.2E+03	10	10	1	0.001	0.999
19		cyclohexane	3.44	2.8E+03	10	10	1	0.000	1.000
20		hexane	4.00	1.0E+04	10	10	1	0.000	1.000
21		octane	5.15	1.4E+05	10	10	1	0.000	1.000
22									
23									
24		**Table 2.B.2**							
25		compound	log*P*	K_D'	V_{aq} (mL)	V_{org} (mL)	*n*	$\alpha_{S(aq)}$	$\alpha_{S(org)}$
26		methanol	-0.74	0.18	10	10	1	0.846	0.154
27		ethanol	-0.30	0.50	10	10	1	0.666	0.334
28		acetone	-0.24	0.58	10	10	1	0.635	0.365
29		acetic acid	-0.17	0.68	10	10	1	0.597	0.403

You can see that I placed the easily extracted solutes in Table 1. These solutes are non-polar and extract to octanol with > 99 % efficiency with only one extraction for an equal volume of organic phase. The solutes in Table 2, being rather polar, are more difficult to extract from water to an organic solvent. Extraction efficiencies are less than 50 %, and obtaining quantitative extractions will require larger volumes of organic solvent, sequential extractions, or both. The next two worksheets illustrated these calculations.

2.C multiple-extractions

This worksheet has a copy of Table 2.B.2 from worksheet 2.B. The exercise is to determine the number of sequential extractions that are necessary to extract > 95 % and > 99 % of each solute from the aqueous phase to the organic phase.

Concepts	Concept Synopsis
extraction efficiency	A calculation to predict the fraction of a solute remaining in the aqueous phase when extracted with an immiscible organic solvent. The expression is given above in question 2 of worksheet 2.B.

Determine the number of sequential extractions that are necessary to achieve the following efficiency.

1. Extract > 95 % of each solute to the organic phase.

First copy your formulas from worksheet 2.B into this table. If you did not move Table 2.B.2, the cells to highlight are H16:I16. Press Ctrl-C, highlight cells H17:I20 in Table 2.C.1, and press Ctrl-V to paste.

The easiest approach to complete this exercise is to simply increase *n* in the table until you achieve 0.95 for $\alpha_{S(org)}$ for each solute. Note that *n* must be an integer – you can't do a half of an extraction. When you are done, your table will look something like the following:

	A	B	C	D	E	F	G	H	I
14									
15		**Table 2.C.1 Extract > 95 %**							
16		compound	log*P*	K_D'	V_{aq} (mL)	V_{org} (mL)	*n*	$\alpha_{S(aq)}$	$\alpha_{S(org)}$
17		methanol	-0.74	0.18	10	10	18	0.049	0.951
18		ethanol	-0.30	0.50	10	10	8	0.039	0.961
19		acetone	-0.24	0.58	10	10	7	0.042	0.958
20		acetic acid	-0.17	0.68	10	10	6	0.045	0.955

Determine the number of sequential extractions that are necessary to achieve the following efficiency:

2. Extract > 99 % of each solute to the organic phase.

This exercise is the same as above, but with the more stringent extraction criterion. We'll simply repeat what we did in question 1. Highlight the content of Table 2.C.1 and press Ctrl-C. Click on a new cell below the existing table and press Ctrl-V to paste a new copy of the table, labeled 2.C.2. Check that the values did not change on copying the formulas. Now change the values of *n* in the table until you achieve 0.99 for $\alpha_{S(org)}$ for each solute. The final results will look like the following figure:

	A	B	C	D	E	F	G	H	I
22									
23		**Table 2.C.2 Extract > 99 %**							
24		compound	log*P*	K_D'	V_{aq} (mL)	V_{org} (mL)	*n*	$\alpha_{S(aq)}$	$\alpha_{S(org)}$
25		methanol	-0.74	0.182	10	10	28	0.009	0.991
26		ethanol	-0.30	0.50	10	10	12	0.008	0.992
27		acetone	-0.24	0.58	10	10	11	0.007	0.993
28		acetic acid	-0.17	0.68	10	10	9	0.010	0.990

2.D extractant-volume

Table 2.D.1 contains example calculations for different conditions to extract acetic acid from water with diethyl ether. You want to determine if changing the extraction solvent to 1-octanol can reduce *n* and the amount of solvent. Determine optimal conditions for this extraction using 1-octanol in place of diethyl ether.

Concepts	**Concept Synopsis**
extraction efficiency	A calculation to predict the fraction of a solute remaining in the aqueous phase when extracted with an immiscible organic solvent. The expression is given in question 2 for worksheet 2.B.
optimum conditions	Choosing extraction conditions to minimize *n* and extractant volume. Minimizing *n* reduces sample prep time. Minimizing $V_{S(org)}$ reduces dilution of the analyte and can reduce subsequent prep time if the extract must be concentrated.

1. Set up Table 2.D.2 for 1-octanol similar to the one for diethyl ether.

A portion of Table 2D.1 is shown below. The highlighted cells meet the extraction criterion of 99.5 % extraction. Diethyl ether will extract acetic acid from aqueous solution, but with either large total volumes or many sequential extractions. The formula for $\alpha_{S(aq)}$ and $\alpha_{S(org)}$ in cell D32 and E32 are:

= (1/(1+$D$26*B32/$F$26))^C32

= 1-D32

The formula for the total volume or extractant is the volume of organic solvent multiplied by *n*:

=B32*C32

	A	B	C	D	E	F	G
23							
24		**acetic acid distribution constants**					
25		organic phase	log*P*	K_D'		V_{aq} (mL)	
26		diethyl ether	-0.36	0.44		10	mL
27		1-octanol	-0.17	0.68			
28							
29							
30		**Table 2.D.1**	**diethyl ether**				
31		V_{org} (mL)	*n*	$\alpha_{S(aq)}$	$\alpha_{S(org)}$	V_{total} (mL)	
32		100	3	0.006	0.994	300	
33		105	3	0.006	0.994	315	
34		110	3	0.005	0.995	330	
35							
36		60	4	0.006	0.994	240	
37		65	4	0.005	0.995	260	
38		70	4	0.004	0.996	280	
39							

An empty Table 2.D.2 for 1-octanol is started in row 61. You can copy cells B32:F32 to row 63 to get started. V_{aq} is the same in all cases, so the $F$26 cell reference remains correct. The only change that is necessary is to replace the K_D' value for diethyl ether with the value for 1-octanol. Change $D$26 to $D$27 in the formula that you copied. The formula in cell D63 should now be:

=(1/(1+$D$27*B63/$F$26))^C63

Highlight cells B63:F63 and copy to cells B64:F65. Leave *n* = 3 and change the values of V_{org} in cells B63:B65 to find the extraction conditions that meet the 99.5 % extraction criterion. By trial and error I find that 70 mL of 1-octanol will meet the 99.5 % criterion. Your result might be slightly different, depending on how finely you increment V_{org}. After highlighting the 99.5 % result, my table looks like the following figure:

	A	B	C	D	E	F
61		**Table 2.D.2**	**1-octanol**			
62		V_{org} (mL)	*n*	$\alpha_{S(aq)}$	$\alpha_{S(org)}$	V_{total} (mL)
63		60	3	0.008	0.992	180
64		65	3	0.006	0.994	195
65		70	3	0.005	0.995	210

Again Highlight cells B63:F63 and now copy to cells B67:F69. Change *n* to 4 for all three cases. Again change the values of V_{org} to find the extraction conditions that meet the 99.5 % extraction criterion. For this case I find 40 mL to be sufficient.

Repeat this process to obtain data for *n* = 5 to *n* = 10. My results look like the following figure. Your results might be slightly different, depending on how finely you increment V_{org}.

	A	B	C	D	E	F
61		**Table 2.D.2**	**1-octanol**			
62		V_{org} (mL)	*n*	$\alpha_{S(aq)}$	$\alpha_{S(org)}$	V_{total} (mL)
63		60	3	0.008	0.992	180
64		65	3	0.006	0.994	195
65		70	3	0.005	0.995	210
66						
67		35	4	0.008	0.992	140
68		40	4	0.005	0.995	160
69		45	4	0.004	0.996	180
70						
71		25	5	0.007	0.993	125
72		28	5	0.005	0.995	140
73						
74		20	6	0.006	0.994	120
75		21	6	0.005	0.995	126
76						
77		16	7	0.006	0.994	112
78		17	7	0.005	0.995	119
79						
80		13	8	0.006	0.994	104
81		14	8	0.005	0.995	112
82						
83		11	9	0.007	0.993	99
84		12	9	0.005	0.995	108
85						
86		10	10	0.006	0.994	100
87		11	10	0.004	0.996	110

2. Determine the optimal conditions from the results.

This exercise is a little open-ended. In most cases, replacing diethyl ether with 1-octanol reduces the total amount of extracting organic solvent by ≈ 1/3. Doing 5 extractions with 1-octanol meets the criteria of achieving 99.5 % extraction in a total organic volume of 140 mL. If I'm doing these extractions myself, 5 extractions is enough. However,

if I have an automated extractor, I will program it to do 9 sequential extractions with 12 mL of the organic phase to decrease the total extractant volume even further.

What I've not considered in answering this question is the suitability of 1-octanol for subsequent steps in the sample processing or analysis procedures. If there are further steps with which 1-octanol is compatible, then it makes sense to change the extracting solvent. However, diethyl ether has the advantage of having a very low boiling point. It is therefore easy to evaporate diethyl ether to concentrate the analyte. The low boiling point also makes it a good solvent for gas chromatography analysis. Whether diethyl ether or 1-octanol is the best extraction solvent will depend on the next steps in the sample preparation and analysis.

2.E percent-recovery (with internal standard calibration)

This worksheet contains:

- a sample preparation procedure for norepinephrine and epinephrine (stress hormones) in blood plasma,
- calibration data for dihydroxybenzylamine, the internal standard added during the solid-phase extraction (SPE) step, and
- analytical measurements for the analytes and internal standard.

The exercise is to calculate the percent recovery of the internal standard and the concentrations of the two analytes.

Concepts	Concept Synopsis
solid-phase extraction	A liquid/solid separation method to "clean up" samples before analysis.
percent recovery	The fraction, expressed in percentage, of the analyte that is recovered for measurement after sample preparation steps. The desired percent recovery is 100 %.
internal standard calibration	An internal standard is a substance that (a) is chemically similar to the analyte and (b) is not present in the sample. A known amount is added to samples before measurement. It serves to calibrate the analytical measurement and can be used to determine the percent recovery after sample preparation procedures.

The purpose of the steps in the SPE procedure are as follows:

- condition the column with 0.5 M HCl followed by water
- add sample (diluted blood plasma)
- wash with water to remove interferences
- add internal standard
- collect analytes and internal standard by eluting with 0.2 M perchloric acid

The analytes and internal standard contain an amine group, and therefore have a positive charge in neutral solution. The SPE stationary phase has a negative carboxylate group, so the positively charged solutes are attracted and held to the stationary phase. Neutral and weakly bound impurities are washed away with water. In the last step of the SPE procedure, the strong acid protonates the carboxylic acid and releases the solutes from the column. After this clean-up procedure, the analytes and internal standard in the eluate are determined by an analytical method.

The goal of a sample preparation step is to isolate analytes from interferences without analyte loss. Adding the internal standard in the SPE procedure provides a check that it and the analytes are eluted completely from the column. The internal standard is chemically similar to the analytes so that it serves as a model of how the analytes behave in the procedure. It also provides a calibration of the analytical measurement. The measurement assumes, or was validated in prior experiments, that the instrument response is the same for the internal standard and the analytes. Other details of the chromatographic analysis are not necessary for the calculation.

1. Determine the percent recovery of the internal standard.

We know that we added 400 pg of the dihydroxybenzylamine internal standard to the sample solution during the SPE procedure. During the subsequent measurement, this amount of dihydroxybenzylamine produced a signal of 179 pA. We have calibration data for dihydroxybenzylamine to relate the measurement signal to concentration. Using this calibration data and the 179 pA measurement, we can determine the amount and the fraction recovered for the internal standard.

Highlight the calibration data and enter a scatter chart (see figure to right). Using LINEST, I find the uncertainty in the intercept to be larger than the value. I will repeat the LINEST function, forcing the intercept through zero. The linear equation for the calibration data is thus:

y = 0.4683x



If you use only the trendline to find the linear equation, enter the slope of 0.4683 somewhere on the worksheet (E34 for my worksheet, see next figure). I've also typed the concentration of the internal standard into cell D40. Note that all concentrations are 'per 500 µL' to match the eluate volume in the SPE procedure.

	A	B	C	D	E	F
32		Dihydroxybenzylamine Calibration Data:				
33			conc (pg/500 µL)	signal (pA)	LINEST	
34			0	2.5	0.4683	0
35			125	51.3	0.011155901	#N/A
36			250	124.8	0.998300447	6.390
37			500	232.1	1762.1703	3
38					71960.8805	122.5095238
39						
40		internal std:	400	pg		

For the measurement in cell D44 and the calibration slope in cell E34, the formula for the concentration of the internal standard is:

=D44/E34

I type this in cell E44. Percent recovery is this concentration divided by the known concentration times 100 %. In cell F44 type:

=(E44/$C$40)*100

where cell C40 holds the concentration of the internal standard. The result is:

	A	B	C	D	E	F	G
40		internal std:	400	pg			
41							
42		Measurement Results:					
43			analyte	signal (pA)	conc (pg/500 µL)	percent recovery	
44			internal std	179	382	95.6	%

2. Determine the concentration of the two unknowns in the plasma sample.

The response or sensitivity of the analytical measurement is the same for the analytes and the internal standard. The internal standard measurement therefore provides a calibration for the analytes. Finding an analyte concentration is a simple proportionality:

$$\frac{\text{unknown concentration}}{\text{unknown measurement}} = \frac{\text{internal standard concentration}}{\text{internal standard measurement}}$$

$$\text{unknown concentration} = \frac{\text{internal standard concentration}}{\text{internal standard measurement}} \text{ unknown measurement}$$

In cells E45 and E46 type:

=($C$40/$D$44)*D45

=($C$40/$D$44)*D46

I use the known internal standard concentration of 400 pg/500 µL from cell C40 in this calculation. I am assuming that the loss of the analytes in the SPE procedure is the same as the loss of the internal standard. Making such assumptions is valid based on prior experiments to confirm that the internal standard and the analytes do behave the same. Since there was a factor of two dilution in preparing the plasma sample, I make a new table for plasma concentration. In cells E49 and E50 I multiply the norepinephrine and epinephrine results by two. The final results will look something like the following figure:

	A	B	C	D	E	F	G
42		Measurement Results:					
43			analyte	signal (pA)	conc (pg/500 µL)	percent recovery	
44			internal std	179	382	95.6	%
45			norepinephrine	177	396		
46			epinephrine	83.5	187		
47							
48					plasma conc (pg/mL)		
49				norepinephrine	791		
50				epinephrine	373		
51							